Alle Formulare Berichte Makros und Module auslesen
Versionen: Access 2007, 2003, 2002/XP und 2000

 FRAGE:  Um alle Tabellen und Abfragen in einer Datenbank in einer Schleife zu bearbeiten, kann bekanntlich der Weg über "TableDefs" oder "QueryDefs" genommen werden:

  strAusgabe = ""
  For Each td In TableDefs 
    strAusgabe = strAusgabe & td.Name & vbCrLf 
  Next td
  MsgBox strAusgabe

  strAusgabe = ""
  For Each qd In QueryDefs 
    strAusgabe = strAusgabe & qd.Name & vbCrLf 
  Next qd
  MsgBox strAusgabe

Wie kann man eine vergleichbare Schleife für Formulare, Berichte, Makros und Module programmieren, für die ja keine geeigneten "Defs"-Auflistungen existieren? 
W. Puscz

 ANTWORT:  Diese Informationen finden Sie in den "Containers"- bzw. "Documents"-Auflistungen des "Database"-Objektes:

  Dim db As DAO.Database
  Dim cnt As DAO.Container
  Dim doc As DAO.Document

  Set db = CurrentDb()
  Set cnt = db.Containers("Forms")
  For Each doc In cnt.Documents
    Debug.Print doc.Name
  Next

  Set doc = Nothing
  Set cnt = Nothing
  Set db = Nothing

Um die anderen Objektarten "Berichte", "Makros" oder "Module" auszulesen, ersetzen Sie 'db.Containers("Forms")' durch 'db.Containers("Reports")', 'db.Containers("Scripts")' und 'db.Containers("Modules")'.

Alle verfügbaren Container listet folgende Routine auf:

  Dim db As DAO.Database
  Dim cnt As DAO.Container

  Set db = CurrentDb()
  For Each cnt In db.Containers
    Debug.Print cnt.Name
  Next

  Set cnt = Nothing
  Set db = Nothing

Wichtig: Der Container "Tables" enthält Tabellen- UND Abfragedefinitionen! Zur klaren Trennung für das Auslesen der Tabellen- und Abfragedefinitionen sollte daher immer die Lösung über TableDefs und QueryDefs verwendet werden.

Diese Techniken können mit allen Versionen genutzt werden. Ab Access 2002/XP stehen alternativ spezielle Auflistungen zur Verfügung, die Sie für das Auslesen beispielsweise von Formularen wie folgt nutzen können:

  Dim db As Object
  Dim obj As AccessObject

  Set db = Application.CurrentProject
  For Each obj In db.AllForms
    Debug.Print obj.Name
  Next

  Set db = Nothing

Analog zu "AllForms" gibt es noch die Auflistungen "AllReports" für Berichte, "AllMacros" für Makros und "AllModules" für Module zur Verfügung.
